

Paper 119-27

An Useful Chart to Display Adverse Event Occurrences in Clinical Trials

Chuanchieh Hsu, Novartis Pharmaceuticals Co., East Hanover, NJ

Zhongwei Zhou, Novartis Pharmaceuticals Co., East Hanover, NJ

J. Michael Hardin, University of Alabama, Tuscaloosa, AL

ABSTRACT

Adverse events (AE) are always important safety issues in clinical trials. It is often an interesting as well as difficult issue for a researcher to determine how best to view a particular AEs in terms of frequency and the time when they occur after randomization across a group of patients or a group of patients compared to another group. A simple and informative graphic to assist the researchers in visualizing and understanding the mechanism of a particular AE is desirable. This paper illustrates how to use SAS/BASE®, SAS Macro, and SAS/Graph® to generate such an AE chart, one that shows the time each AE occurs and, more importantly, the severity of each corresponding AE over the duration of a clinical trial. Summaries of AE counts are also presented in the graph.

Keywords: Adverse event, severity, graphical visualization, clinical trials.

INTRODUCTION

Safety and efficacy reports are the most important components in a clinical study report. Statisticians and clinicians review both reports in drawing conclusions on the results of a study. Usually statisticians and programmers make several graphics to display efficacy endpoints and to display the treatment effect. Efficacy parameters are usually the primary parameters. Safety issues, on the other hand, are usually secondary endpoints. Consequently, less attention is often given to safety endpoints, such as adverse events (AEs). Statisticians and clinicians usually conclude safety issues based solely on the safety table generated by programmers.

Graphical displays are very helpful to present either large quantities of information or complicated results for quick and easy interpretations. The purpose of this paper is to present a clear and useful way to indicate the time and the severity of a particular AE for each subject for a typical study examining differences between a placebo and treatment groups. Such a graphical representation provides an easy visualization for clinicians to see when and how many AE occurs in each patient and the severity of each occurrence.

A SAS program was developed for this purpose using SAS/BASE®, SAS Macro, and SAS/GRAPH®.

PREPARING THE DATA

The example used in this paper is from AE data of a fictitious clinical study with three groups, one placebo and two treatments. It is assumed that each patient is asked to daily record the presence of headache (one of the AEs) and its severity in a diary. At the end of the study, all dairies are collected and entered into a database, such as Oracle, Sybase, or MS/SQL server.

For convenience, we assume the data have been cleaned and prepared for the final safety report. The data used in this paper contains the following variables: treatment group, patient number, AE duration, and the severity. The AE duration is the time of an AE occurrence from that patient's entry into the study (usually the date of randomization of the study).

```
*** Date input from an ASCII file, and the program to read it.
***,
data all;
  input rx patno $ dur sev;
  cards;
```

```
3      10309      6      1
3      10309      78     2
3      10309      82     1
1      20609      27     2
1      20609      10     2
1      20609      16     1
1      20609      48     1
2      30201      8      1
2      30201      9      2
...
;;;
run;

proc format;  value sev 1 = "Mild"  2 = "Moderate"  3 =
"Severe"; run;
```

THE USUAL PRESENTATION

In the pharmaceutical industry, the AEs are summarized in safety tables generated by programmers. A typical format for this summary is shown below.

```
Safety table 1.1 (Page 1 of 3)

Adverse events (Safety patients)

Headache
  Placebo   Trt One   Trt Two   Total
  (N=98)    (N=96)    (N=99)    (N=293)
  n (%)     n (%)     n (%)     n (%)
-----
  8 (8.2)   13 (13.5)  11 (11.1)  32 (10.9)
  ...
```

Same AE's are only counted once and the worst AE is used.

This safety table shows the differences among the groups. The occurrences of adverse experiences (AEs) within the treatment groups are typically analysed using Fisher's exact tests, and then summarized by randomized treatment group as in the above table. The purpose of the table is to present a summary of the occurrences of the AEs and the difference among the treatment groups. However, it does not portray the magnitude of this particular AE in terms of when and how severe the occurrence. It does not show how long this AE occurred, how often, how many, nor how severe the occurrence throughout the course of the study. The following chart, on the other way, helps statisticians and clinicians to visualise, for each patient, how many, when and how severe AEs occur.

TO PREPARE THE GRAPH

Since the program is mainly written in a SAS Macro, several macro variables need to be preset prior to running the program. For example,

```
%let safevar=sev; * set safety variable;
%let rx=rx; * set treatment variable;
%let numrx=3; * set num of treatment groups;
%let safefmt=sev; * set format variable;
%let numdays=100; * max # of days in med;
%let colors=blue purple red; * set colors;
```

The first part of the program reads the AE data into SAS. The data are then sorted by treatment group and patient number with the option, **nodupkey**, in order to count the total number of patients who have the AE in each group. Of course there are other ways to do the same task. Titles can be added to the plot if needed.

```
data titles(keep=xsys ysys hsys color function
x y text position size);
  length position xsys ysys hsys $1 color
  function $8 text $200;
  retain %do i=1 %to &numttl; ttl&i "&&ttl&i"
  %end; color 'cyan' xsys ysys hsys '3'
  position '5' function 'label' size
  &tsize x &xttl y;
  array ttls{&numttl} $200 ttl1-ttl&numttl;
  do i=1 to &numttl;
    y=&yttl-(i-1)*(size+0.2); text=ttls{i};
    output;
  end;
run;
```

To separate the titles and the plot, two thick lines are then added to the plot for better presentation.

```
data brkline;
  length xsys ysys hsys $1 function color $8;
  retain xsys ysys hsys '3' function y size x
  color;
  size=1; y=81; color='cyan';
  function='move'; x=2; output;
  function='draw'; x=98; output;
  size=0.5; y=80; color='purple';
  function='move'; x=2; output;
  function='draw'; x=98; output;
run;
```

The main part of the program is given below. It uses SAS/Base and SAS/Graph annotation for the graph. The first step of making this AE plot is to show the name, number of patients who have the AEs, and the total number of patients in the treatment group (data **SUBTTL**). Then the dataset **OCCUR** is the next step to draw the line from the day 0 to the end of the study for the patient and draw a symbol (circle, square, or star) in different color to present the severity of the AE occur on that day. Legend underneath the plot is then being setting up to indicate the symbol in the plot. The data set, **ANNO**, for the annotation of this plot is created by merging **OCCUR** and **SUBTTL**. Finally, Proc GPLOT and Proc GREPLAY are used to make the AE plot with annotation, legend, and the titles. Note that patient numbers are shown on the y-axis for each treatment group in the plot.

```
* The settings here are for the plot *;
%let gstart=12;
%let xlength=22;
%let xinterv=6;
%let stsize=2;
data subtll(drop=aetot totnum);
  length position xsys ysys hsys $1 function $8 text $200;
  retain xsys ysys hsys '3' position '5' function 'label'
  size &stsize y 78 x;
  merge totnum numsub;
  by &rx;
  x=&gstart+(&rx-1)*(&xinterv+&xlength)+&xlength*0.5;
  text=trim(left(put(&rx, &&rxfmt.)))||' ('||
  trim(left(put(aetot, 8.)))||' '||
  trim(left(put(totnum, 8.)))||')';
run;
```

```
proc sort data=all out=all;
  by &rx &patno;
run;

data occur;
  merge all numsub;
  by &rx;
run;

data occur;
  length xsys ysys hsys $1 color function $8 text $200;
  format color $char8.;
  retain hsys '3' xsys ysys '2' function x y size style
  line color seq;
  set occur;
  by &rx &patno notsorted;
  if first.&rx then seq=0;
  if first.&patno then do;
    seq+1; size=0.1; line=1; y=seq;
    function='move'; x=0; output;
    function='draw'; x=&numdays; output;
  end;
  function='label'; size=2; x=&dur;
  y=seq+aetot*0.01;
  text='O'; style='swiss';
  if sev=1 then color="&color1";
  else if sev=2 then color="&color2";
  else if sev=3 then color="&color3";
  output; color='black'; style="";
run;
```

Here is the preparation for the legend of this plot.

```
%let ylgnd=5;
%let lgndsz=1.8;

data legend;
  length xsys ysys hsys position $1 color function $8
  text $200;
  format color $char8.;
  retain hsys xsys ysys '3' function 'label'
  position '6' y &ylgnd
  size &lgndsz x color;
  x=30; text='O'; color="&color1"; output;
  x=32; text=put(1, &safefmt.); color="black"; output;
  x=45; text='O'; color="&color2"; output;
  x=47; text=put(2, &safefmt.); color="black"; output;
  x=60; text='O'; color="&color3"; output;
  x=62; text=put(3, &safefmt.); color="black"; output;
run;
```

MAKING THE PLOT

The main body of this AE plot is to have the patient who has AEs occurred in the study, and then draw a line to present the duration of that patient. And then draw different symbols in different colors to present the severity of each AE occurs to that patient.

```
Data anno;
  set subtll occur;
  by &rx;
run;

data anno2;
  set %if &numttl gt 0 %then %str(titles); brkline legend;
run;
```

```

proc gslide gout=aepplot annotate=anno2 name="fig0";
run;

%let minx=0;
%let maxx=&numdays;
%let byx=20;
%let xvalsz=1.5;
%let yvalsz=1.5;
%let xlabel=;
%let xlblsz=2;
%let ylabel=;
%let ylblsz=2.5;

data plot;
  x=.; y=.;
run;

proc sort data=occur(keep=&patno &rx seq) out=pat
nodupkey;
  by &rx seq;
run;

%let ylength=60;

%do i=1 %to &numrx;
  %let x0=%eval(&gstart+(&i-1)*(&xinterv+&xlength));
  %let y0=15;

  axis1 order=(0 to &maxx by &byx)
    origin=(&x0, &y0) width=2 length=&xlength
    major=(height=0.5 width=2) minor=none
    value=(h=&xvalsz)
    %if %quote(&xlabel) eq %then %str(label=none);
    %else %do; label=(height=&xlblsz "&xlabel") %end;;

  data pat2;
    set pat(where=(&rx=&i));
    by &rx seq;
  run;

  data _null_;
    set pat2 end=eof;
    call symput('patnum'||left(seq), trim(left(&patno)));
    if eof then call symput('maxy', trim(left(put(seq, 8)))));
  run;

  axis2 order=(1 to &maxy by 1) origin=(&x0, &y0) width=2
    length=&ylength major=(height=0.5 width=2)
    minor=none
    value=(h=&yvalsz %do x=1 %to &maxy; tick=&x j=
      "&patnum&x" %end;)
    %if (%quote(&ylabel) ne ) &i=1 %then
    %do; label=(height=&ylblsz angle=90 "&ylabel")
    %end;
    %else %str(label=none);

  proc gplot data=plot gout=aepplot;
    plot y*x /haxis=axis1 vaxis=axis2 href=3 lhref=3
      nolegend frame
      annotate=anno(where=(&rx=&i)) name="fig&i";
  run;
%end; ** End of RX **;

```

Now, it is time to put titles, subtitles, footnotes, and legends together using Proc GREPLAY and output a CGM file of the graph.

```

Filename gsf "aepplot.cgm";


Goptions reset=all ftext=swiss gunit=pct
gsfmode=replace
          device=cgmof971 ctext=black
ctitle=black
          gsfname=gsf;

proc greplay nofs igout=aepplot;
  tc=temp;
  tdef a10 1/1lx=0 ulx=0 urx=100 lrx=100
  lly=0 uly=100
          ury=100 lry=0;
  template a10;
  treplay 1: "fig0" %do I=1 %to &numrx; 1:
  "fig&i" %end;;
run;

```

Finally, the AE plot is complete and shown in Figure 1.

Figure 1



CONCLUSION

The conclusion summarizes your paper and ties together any loose ends. You can use the conclusion to make any final points such as recommendations predictions, or judgments. It is hard to get a clear picture to know when adverse events occur and how severe if it occurs after taking a medication. The AE chart showed in this paper provides a clear and easy visual presentation for investigators to visualize the differences between groups for that particular AE. Furthermore, the chart can be extended. For instance, if there is titration steps for the medication, thus it is very clear to see if more events of the AE

occur after taking a higher dose. Statistical analysis and p-values for the comparison of the groups can be added to the plot as well.

The codes and explanations presented in this paper are only one of many ways the graph could have been produced. Modifications to the graph, such as titles, footnotes, fonts, p-value, and graph output format, can easily be made by SAS programmers. SAS/GRAPH® software offers the programmers many creative ways to meet these particular needs.

REFERENCES

SAS Institute Inc., SAS/GRAPH software, Volumes 1 & 2: Reference Version 6, First Edition, Cary NC: SAS Institute Inc., 1990.

SAS Institute Inc., SAS/STAT User's Guide, Volumes 1 & 2, Version 6, Fourth Edition, Cary NC: SAS Institute Inc., 1990.

SAS Institute Inc., SAS Guide to Macro Processing, Version 6, Second Edition, Cary NC: SAS Institute Inc., 1990.

SAS Institute Inc., SAS/BASE User's Guide, Version 6, Fourth Edition, Cary NC: SAS Institute Inc., 1990.

CONTACT INFORMATION

Chuanchieh Hsu, Ph.D. 
Senior Statistician, .Novartis Pharmaceuticals Co.
JAY.HSU@PHARMA.NOVARTIS.COM

Zhongwei Zhou, 
Scientific Analyst, Novartis Pharmaceuticals Co.
ZHONGWEI.ZHOU@ PHARMA.NOVARTIS.COM

J. Michael Hardin, Ph.D.
Professor, Dept. of Information System, Statistics and
Management Science
University of Alabama.
mhardin@cba.ua.edu